Welcome and thank you for joining. The webinar will begin shortly.
## Welcome

| QUESTIONS & TECHNICAL SUPPORT | Participants are on mute.  
|                              | Use the **Chat Box** to ask questions or request support.  
<table>
<thead>
<tr>
<th></th>
<th>Questions will be addressed during designated Q&amp;A periods.</th>
</tr>
</thead>
</table>
| CLOSED CAPTIONING SERVICE     | Closed captioning is available.  
|                              | A link to the service will be provided in the Chat Box.       |
| WEBINAR RECORDING & MATERIALS | The webinar is being recorded.  
|                              | Materials will be posted online in approximately three weeks.  
|                              | Email notification will be sent.                           |
Speakers

Carolyn Reyes-Guzman, PhD, MPH
*Program Director*
National Cancer Institute
Behavioral Research Program
Tobacco Control Research Branch

Elizabeth Seaman, PhD, MHS
*Project Manager*
CDC Foundation

Kelvin Choi, PhD, MPH
*Senior Investigator*
National Institute on Minority Health and Health Disparities
Division of Intramural Research
How to Conduct Analyses Using the TUS-CPS 1992-2019 Harmonized Dataset

Carolyn Reyes-Guzman, PhD, MPH
NCI, Tobacco Control Research Branch
Webinar Agenda

1. Background on data harmonization
2. Self-response and replicate weights for variance estimation
3. How to run analyses using harmonized data
4. Applied examples
   - Disparities in menthol cigarette use
   - Tobacco use surveillance among small populations

Questions?

Please type into the chat box and moderators will ask them at the end of the presentation
Disclaimer

The views and opinions expressed are the presenters’ and do not necessarily represent the views, official policy, or position of the U.S. Government, U.S. Department of Health and Human Services or any of its affiliated institutions or agencies.
Background on Data Harmonization
TUS-CPS Background

- NCI has sponsored this survey of tobacco use administered as part of the US Census Bureau’s Current Population Survey (CPS) every 3-4 years since 1992-93, with co-sponsorship from the FDA since 2014 and the CDC from 2001-2007.
- Most recent publicly released data are from the 2018-2019 wave; 2022-2023 wave currently being planned.
- CDC was a co-sponsor with NCI from 2001-2002 through 2006-2007
What is TUS useful for?

- TUS data can be used by researchers to:
  - Monitor tobacco control progress and assess long-term cross-sectional population trends;
  - Track tobacco health disparities;
  - Evaluate tobacco control programs; and
  - Examine national, state and county*-level data.

- Data from the TUS can be linked to other CPS supplements, including detailed labor force, occupational, and economic and health insurance data (ASEC or March Supplement) that can be related to tobacco use, as well as to mortality data from the National Death Index.

*where available in these smaller geographical unit areas
Effort behind data harmonization

- Researchers who previously wanted to examine long-term trends had to track variable names for a specific question of interest over several survey waves.

- Harmonized file:
  - one variable name across all waves, using “flag” variable to track survey year
  - Adult self-respondents, no proxy responses
  - variables with >1 wave of data
Inclusion/exclusion criteria for harmonization

- Variables to harmonize as-is:
  - remained consistent over time, harmonized in their current state

- Variables to harmonize after adjustments:
  - slight wording changes over time or those for which universe had slight variations across survey waves were first adjusted

- Variables to drop:
  - items with significant wording, structure, or universe variations

- Data dictionary + crosswalk available on TUS-CPS site

- Harmonized file user guide to be released Fall 2021 (with examples of SAS/SUDAAN code)
Harmonized Variables

- Core variables from CPS (sociodemographic, geographic, occupational, economic)
- Cigarette use – including menthol cigarettes since 2003
- Workplace and home smoking restrictions
- Attitudes toward smoke-free policies in indoor work areas/public places
- Smoke-free attitudes for multi-unit housing*

- Advice by physician/dentist
- Health perceptions/beliefs (harm reduction)
- Smoking history, cessation, former smokers
- Use of other tobacco products: cigar, pipe, smokeless tobacco use, e-cigarettes
- Attempts to quit smoking by switching to e-cigarettes*
- Flavored tobacco products

*Only include 2014-2015 and 2018-2019 TUS waves
Self-Response and Replicate Weights
Dealing With TUS-CPS Weights

- The harmonized dataset includes a full-sample weight for self respondents (SmplWgt)
- “Full sample weights are created to compensate for differential selection probabilities, nonresponse, and under-coverage of the target population of U.S. adults”\(^1\)
- But there is a second type of weights available for the TUS-CPS too –
  - “Replicate weights, which can accommodate various types of statistical analyses, are created to more accurately estimate standard errors by accounting for the complex survey design”\(^1\)
- Depending on the goals of your analysis, you may need to use replicate weights

TUS-CPS Replicate Weights

- The number of replicate weights used for TUS-CPS has changed over time, and, accordingly, replicate weights are provided in three separate files:
  - 1992-93: 48 replicate weights
  - 1995-96 through 2003: 80 replicate weights
  - 2006-07 through 2018-19: 160 replicate weights
Linking TUS-CPS Harmonized File to Replicate Weights

- To run analyses, first need to merge TUS respondents with their corresponding replicate weights by ID
- “RecordID” variable in harmonized file is unique within survey year and survey month
- Same RecordID variable was created for replicate weight files
- Files merged by SurYear, SurMonth and RecordID
- SAS code to merge files available on TUS-CPS site
Linking TUS-CPS Harmonized File to Replicate Weights

Data Rep9293;
Infile Rep9293;
Input @001 SurYear 4.
   @005 SurMonth 2.
   @007 RecordID $Char22.
   @029 SmplWgt 11.4
   @040 (RepWt001-RepWt048) (11.4);
Run;

Data Rep9503;
Infile Rep9503;
Input @001 SurYear 4.
   @005 SurMonth 2.
   @007 RecordID $Char22.
   @029 SmplWgt 11.4
   @040 (RepWt001-RepWt080) (11.4);
Run;

Data Rep0619;
Infile Rep0619;
Input @001 SurYear 4.
   @005 SurMonth 2.
   @007 RecordID $Char22.
   @029 SmplWgt 11.4
   @040 (RepWt001-RepWt160) (11.4);
Run;

Data Reps;
Set Rep9293 Rep9503 Rep0619;
Run;

Proc Sort Data=Reps;
   By SurYear SurMonth RecordID;
Run;

Proc Sort Data=Harmon;
   By SurYear SurMonth RecordID;
Run;

Data Harmon;
   Merge Harmon(In=In1)
      Reps(In=In2);
   By SurYear SurMonth RecordID;
   Harm_File=In1;
   Reps_File=In2;
Run;
Using TUS-CPS Replicate Weights

- Fay’s method (a variation of balanced repeated replication, or BRR) when conducting variance estimation for the TUS-CPS with replicate weights
  - For frequency tables using **one wave** of data:
    - In SAS: use `Proc SurveyFreq` and `VarMethod=BRR` (Fay=0.5)
    - In SUDAAN: use `Proc Crosstab` and `ADJFay=4`
  - For **multiple waves** of data, use `Fay=0.75`, SAS; `ADJFay=16`, SUDAAN
Using Weights in Harmonized Data Analyses
Example: using 1 wave of data

- *Calculate mean cigarettes per day among overall current smokers during the 2014-2015 wave (self-respondents only)*

- From the TUS-CPS website, download in the ZIP file containing the 1992-2019 data (.dat file), a SAS program to read in the data, a SAS program with formats for harmonized variables, three files of replicate weights (.dat files), and three corresponding SAS programs to read in the replicate weight data

- Open the SAS program to read in the self-response replicate weight files and merge them with the survey data (located in the Harmonized TUS-CPS dataset .zip file available on the TUS-CPS website). Modify the Filename, Libname and Include statements to match location and names of the data files and the format program, and run the program.
Example: using 1 wave of data

- From the dataset, select the TUS-CPS self-respondents who completed the 2014-2015 survey wave.
  
  ```
  Data harmon1415;
  Set harmon;
  If SURWAVE=9; /*2014-2015 Survey Wave*/
  Run;
  ```

- Divide person-weights and replicate weights by 3 (the number of months of data being combined for analysis).
  
  ```
  /* RepWt001-RepWt160 are the replicate weights */
  Data harmon1415;
  Set harmon1415;
  Array Wgts(160) RepWt001-RepWt160;
  Do I = 1 to 160;
  Wgts(I)=Wgts(I)/3;
  End;
  NSmplWgt=SmplWgt/3;
  Run;
  ```
Example: using 1 wave of data

- Recode variables needed for analysis: current cigarette smoking status (yes/no, called CURRSMK), cigarettes smoked per day (continuous, among current smokers, called CIGPD)

```sas
Data harmon1415;
  Set harmon1415;
  /* CurrSmk: Current Cigarette Smoking Status */
  If CigStat in (2,3) Then CurrSmk=1; /* Current Cigarette Smoker */
  Else If CigStat in (1,4) Then CurrSmk=0; /* Non-Smoker */
  Else CurrSmk=.;

  /* CigPD: Number of Cigarettes Per Day */
  If CigStat=2 & (0<=CPDD<=99) Then CigPD=CPDD; /* Daily Smokers */
  Else If CigStat=3 & (0<=CPDS<=30) Then CigPD=CPDS; /* Non-Daily Smokers */
  Else CigPD=.;

  Label CurrSmk = "Current Cigarette Smoking Status"
  CigPD = "Number of Cigarettes Per Day";

  Keep RecordID CurrSmk CigPD NSmplWgt RepWt001-RepWt160;
  Format CurrSmk CurrSmkF.;
Run;
```
Example: using 1 wave of data

- The following SAS code will output weighted means and standard errors for cigarettes smoked per day among current cigarette smoking (CURRSMK=1) adult self-respondents.

```sas
Proc SurveyMeans Data=harmon1415 VarMethod=BRR (Fay=0.5);
Var CigPD;
Domain CurrSmk;
Weight NSmplWgt;
RepWeights RepWt001-RepWt160;
Run;
```
Example: pooling multiple waves of data

Calculate the prevalence of overall current smoking during the 2003 and 2006-2007 waves (self-response)

- From the full, harmonized dataset (survey data and replicate weights), select the TUS-CPS self-respondents who completed the 2003 or 2006-2007 survey waves.

```plaintext
Data harmon030607;
Set harmon;
If SURWAVE in (6,7); /* 2003, 2006-2007 Survey Waves */
Run;
```
Example: pooling multiple waves of data

- Both the person-weights and replicate weights must be divided by 6 (the number of months of data being combined for analysis; 3 months from each survey wave), so that the weights total to the average size of the U.S. population during the 2003 and 2006-2007 time periods.

```
Data harmon030607(Drop=I J RepWt001-RepWt160 SmplWgt);
Set harmon030607;
Array OldR(160) RepWt001-RepWt160;
Array NewR(240) NWgt001-NWgt240;
NSmplWgt=SmplWgt/6;
/* The 2003 dataset has 80 replicate weights and the 2006-07 has 160, so it is necessary to expand or extend the number of replicate weights to 240 so that all survey months have an equal number of replicate weights. */
If SurWave=6 Then Do;
  Do I = 1 to 80;
    /* The following mathematical adjustment factor extends the number of replicate weights from 80 to 240 so that they are equalized with the other survey waves. */
    NewR(I)=(1/6) * (SmplWgt+(.866025*(OldR(I)-SmplWgt))); /* .866025 = 1/2 x (Sqrt(240/80) */
  End;
/* For 2003, 160 replicate weight are generated beyond the first 80 by dividing the main sample weight by the number of surveys. */
  Do I = 81 to 240;
    NewR(I)=SmplWgt/6;
  End;
End;
/* Need to do the reverse for the 2006-07 survey data. */
Else Do;
  Do I = 1 to 80;
    NewR(I)=SmplWgt/6;
  End;
  Do I = 81 to 240;
    J = I - 80;
    NewR(I)=(1/6) * (SmplWgt+(.612372*(OldR(J)-SmplWgt))); /* .612372 = 1/2 x (Sqrt(240/160) */
  End;
End;
Run;
```
Example: pooling multiple waves of data

- Recode current cigarette smoking status (yes/no, called CURRSMK)

```plaintext
Data harmon030607;
  Set harmon030607;

  /* CurrSmk: Current Cigarette Smoking Status */
  If CigStat in (2,3) Then CurrSmk=1; /* Current Cigarette Smoker */
  Else If CigStat in (1,4) Then CurrSmk=0; /* Non-Smoker */
  Else CurrSmk=.;
  Label CurrSmk = "Current Cigarette Smoking Status";

  Keep RecordID CurrSmk NSmplWgt NWgt001-NWgt240;
  Format CurrSmk CurrSmkF.;
  Run;
```

- Output weighted prevalence estimates for current cigarette smoking (CURRSMK) among adult self-respondents.

```plaintext
Proc SurveyFreq Data=harmon030607 VarMethod=BRR (Fay=0.75);
  Tables CurrSmk /CL;
  Weight NSmplWgt;
  Repweights NWgt001-NWgt240;
  Run;
```
Example: pooling the entire harmonized dataset

- Estimate the adjusted odds ratio for the association between demographic factors and smoking status among all TUS-CPS adult self-respondents from 1992-2019

- Both the person-weights and replicate weights must be divided by 29 (the number of months of data being combined for analysis; 3 months from each survey wave), so that the weights total to the average size of the U.S. population during the 1992-2019.
Example: pooling the entire harmonized dataset

```sas
Data harmon_recode(Drop=I J RepWt001-RepWt160 SmplWgt);
    Set harmon_recode;
    Array OldR(160) RepWt001-RepWt160;
    Array NewR(288) NWgt001-NWgt288;
    NSmplWgt=SmplWgt/29;
    /* The 1992-93 data has 48 replicate weights, the 1995-96 through 2003 data has 80 RPs and the 2006-07 onwards has 160 RPs, so it is necessary to expand or extend the number of replicate weights to 288 so that all survey months have an equal number of replicate weights. */

    If SurWave=1 Then Do;
        Do I = 1 to 48;
            NewR(I)=(1/29) * (SmplWgt+(1.224745*(OldR(I)-SmplWgt))); /* 1.224745 = 1/2 x (sqrt(288/48)) */
        End;
        /* For 1992, 48 replicate weight are generated beyond the first 48 by dividing the main sample weight by the number of surveys. */
        Do I = 49 to 288;
            NewR(I)=SmplWgt/29;
        End;
    End;

    Else If SurWave in (2,3,4,5,6) then do;
        Do I = 1 to 48;
            NewR(I) = SmplWgt/29;
        End;
        Do I = 49 to 128;
            J = I-48;
            NewR(I)=(1/29) * (SmplWgt+(0.948683*(OldR(J)-SmplWgt)));
        /* 0.948683 = 1/2 x (sqrt(288/80)) */
        End;
        Do I = 129 to 288;
            NewR(I) = SmplWgt/29;
        End;
    End;

    Else If SurWave in (7,8,9,10) then do;
        Do I = 1 to 128;
            NewR(I)=SmplWgt/29;
        End;
        Do I = 129 to 288;
            J = I - 128;
            NewR(I)=(1/29) * (SmplWgt+(0.670820*(OldR(J)-SmplWgt)));
        /* 0.670820 = 1/2 x (sqrt(288/160)) */
        End;
    End;

Run;
```
Acknowledgements

- Margaret Mayer, PhD, MPH (NCI)
- Todd Gibson (IMS/NCI Contract)
- Nalini Corcy, MHA (BLH/NCI Contract)
- Dana Chomenko, MPH, PMP (BLH/NCI Contract)
- TUS-CPS Team (NCI DCCPS, NCI DCEG, and NIH NIMHD)
- FDA CTP
Questions?

TUS-CPS Website - https://cancercontrol.cancer.gov/brp/tcrb/tus-cps

Analyses Using the TUS-CPS 2003-2015 Harmonized Dataset

Example Using Menthol Cigarette Use Data

Elizabeth Seaman, PhD, MHS

CDC Foundation
1. How To Conduct Analyses and Code Examples
2. Summary of Findings
How to Conduct Weighted Analyses: Example Using Menthol Cigarette Use Data
Research Question: How has menthol cigarette smoking among current smokers changed between 2003 and 2014-2015?

- The TUS-CPS first included questions about menthol cigarette use in the 2003 survey
  - To maximize the amount of data available and get the full picture we wanted to include 4 waves (2003, 2006-2007, 2010-2011 and 2014-2015) of TUS-CPS data
- We wanted to adjust our estimates for demographic factors
Menthol Cigarette Use in TUS-CPS

2003 and 2006-2007

Is your usual cigarette brand menthol or non-menthol?

(1) Menthol
(2) Non-menthol
(3) NO USUAL TYPE

2010-2011 and 2014-2015

Do you usually smoke menthol or non-menthol cigarettes?

(1) Menthol
(2) Non-menthol
(3) NO USUAL TYPE
Menthol Cigarette Use in Harmonized Dataset

<table>
<thead>
<tr>
<th>CIGTYPE</th>
<th>126-127</th>
</tr>
</thead>
<tbody>
<tr>
<td>Type of cigarette (every day and some days smokers, 2003 and later)</td>
<td>-9 = No response</td>
</tr>
<tr>
<td></td>
<td>-3 = Refused</td>
</tr>
<tr>
<td></td>
<td>-2 = Don't know</td>
</tr>
<tr>
<td></td>
<td>-1 = Not in universe</td>
</tr>
<tr>
<td></td>
<td>1 = Menthol</td>
</tr>
<tr>
<td></td>
<td>2 = Non-menthol</td>
</tr>
<tr>
<td></td>
<td>3 = No usual type</td>
</tr>
</tbody>
</table>
Special Consideration – Analyzing Across Waves

- 2003 and earlier surveys had only 80 replicate weights
- 2006-2007, 2010-2011 and 2014-2015 have 160 replicate weights
- We wanted to conduct analyses across years with different numbers of replicate weights, we needed to create 240 adjusted replicate weights (160+80)
Order of Steps

1. Format Harmonized Dataset; Output Data Needed
2. Format and Merge Replicate Weights Needed
3. Merge Harmonized Dataset and Replicate Weights
4. Adjust Replicate Weights
5. ANALYZE DATA!
Merging Harmonized Data to Replicate Weights

/*sort both datasets by ID*/
proc sort data=all.rep; /* this dataset is the replicate weights dataset*/
by ID;
run;

proc sort data=MyLib.harmones2; /* this dataset is the harmonized dataset*/
by ID;
run;

/*merge replicate weights and harmonized dataset*/
data all.TUSmerge;
merge MyLib.harmones2(in=a) all.rep(in=b);
by ID;
if a and b;
run;
Adjusting Replicate Weights

Data TUSmerge2(Drop=I J repwgt1-repwgt160 repwgt0);
Set all.TUSmerge;
Array OldR(160) repwgt1-repwgt160;
Array NewR(240) NWgt001-NWgt240;
NSmplWgt=repwgt0/12;

If SURYEAR=2003 Then Do; /*2003 has 80 rep wgts, we need to create an adjustment to extend the number of wgts to 240*/
   Do I = 1 to 80;
      NewR(I)=(1/12) * (repwgt0+(.866025*(OldR(I)-repwgt0))); /*.866025=1/2 x (Sqrt(240/80)*/
   End;
   Do I = 81 to 240;
      NewR(I)=repwgt0/12; End; End;

Else Do; /*2006-2007, 2010-2011 and 2014-2015 have 160 rep wgts, we need to create an adjustment to extend the number of wgts to 240*/
   Do I = 1 to 80;
      NewR(I)=repwgt0/12; End;
   Do I = 81 to 240;
      J = I - 80;
      NewR(I)=(1/12) * (repwgt0+(.612372*(OldR(J)-repwgt0))); End; /*.612372=1/2 x (Sqrt(240/160)*/
   End;
Run;
Creating data steps for analysis

```sas
Data TUSmerge3;
set TUSmerge2;
if CIGSTAT <1 then CIGSTAT=.;
if CIGTYPE <-1 then CIGTYPE=.;
if SEX <1 then SEX=.;
/*recode cigstat*/
NewCIGSTAT=.;
if CIGSTAT in (1,4) then NewCIGSTAT=0; /*never/former cigarette smokers*/
else if CIGSTAT in (2,3) then NewCIGSTAT=1; /*current every day or some day cigarette smokers*/
/*recode cigtype*/
NewCIGTYPE=.;
if CIGTYPE=1 then NewCIGTYPE=1; /*usual type menthol cigarette smokers*/
else if CIGTYPE=2 then NewCIGTYPE=2; /*usual type non-menthol cigarette smokers*/
else if CIGTYPE=3 then NewCIGTYPE=3; /*no usual type cigarette smokers*/
else if CIGSTAT=1 then NewCIGTYPE=4; /*not current cigarette smokers*/
else if CIGSTAT=4 then NewCIGTYPE=4; /* not current cigarette smokers*/
NewMenthr=.;
if NewCIGTYPE = 1 then NewMenthr = 1; /*menthol*/
else if NewCIGTYPE = 2 then NewMenthr = 0; /*non-menthol*/
/*recode age*/
NewAGE=.;
IF (18 <= AGE <= 24) THEN NewAGE = 1;
else if (25 <= AGE <= 34) THEN NewAGE = 2;
else if (35 <= AGE <= 44) THEN NewAGE = 3;
else if (45 <= AGE <= 54) THEN NewAGE = 4;
else if AGE>= 55 THEN NewAGE = 5;
/*recode self-response weight for 12 data collection points*/
AllSRWEIGHT = SRWEIGHT/12;
-----------continued
```
How to Obtain Adjusted Prevalence Estimates using PREDMARG

```
proc rlogist Data=TUSmergeES3 filetype=sas design=BRR;
Setenv DecWidth=3;
Weight AllSRWEIGHT;
RepWgt NWgt001-NWgt240/ADJFay=16;
subpopx incflag=1;
class NewSURWAVE SEX NewAge NewRACE NewEMPLSTAT METSTAT REGION NewEDU;
model NewMenthr = NewSURWAVE SEX NewAge NewRACE NewEMPLSTAT METSTAT REGION NewEDU;

predmarg NewSURWAVE;
print predmrg="PREDMARG" / pred_mrg=default lowpmfmt=f6.4 uppmfmt=f6.4 predmrgfmt=f8.4 t_prdmrgfmt=f8.2 p_prdmrgfmt=f7.4;
run;
```
Summary of Findings

**Overall weighted adjusted prevalence by TUS-CPS cycle***

<table>
<thead>
<tr>
<th></th>
<th></th>
<th></th>
<th></th>
<th></th>
</tr>
</thead>
<tbody>
<tr>
<td>Weighted %, (95% CI)</td>
<td>27.9 (27.3, 28.5)</td>
<td>28.0 (27.3, 28.7)</td>
<td>31.1 (30.4, 31.8)</td>
<td>33.3 (32.5, 34.1)</td>
</tr>
</tbody>
</table>

Adjusted for sex, age, race, employment status, metropolitan status, region and educational attainment.

Unpublished data; submitted for publication

Unpublished data; submitted for publication
Summary of findings

- In the face of declining cigarette smoking prevalence in the U.S. population between 2003 and 2014-2015, menthol cigarette smoking among current smokers has increased.
- Differences in prevalence of menthol use among current smokers exist for various demographic subgroups.
- Young adults (age 18-24), women, and non-Hispanic Black smokers have higher prevalence of menthol use compared to their counterparts.
- Non-Hispanic Black smokers have nearly double the prevalence of menthol smoking compared to any other racial or ethnic group.
Leveraging Data from the Current Population Survey for Tobacco Use Surveillance among Small Populations

- Kelvin Choi, PhD, MPH
- Senior Investigator
- Head, Social and Behavioral Sciences Section
- Division of Intramural Research
- TUS Data User Webinar Series
- Analyses Using the TUS-CPS 1992-2019 Harmonized Dataset
- July 1, 2021 – 2:00-3:30pm ET
Disclaimer

• Comments and opinions expressed in this presentation are the presenter’s own and do not necessarily represent those of the U.S. Government, Department of Health and Human Services, National Institutes of Health, and National Institute on Minority Health and Health Disparities.

• The word “tobacco” in this presentation refers to commercially manufactured, marketed, and distributed tobacco products.
Variables from the Tobacco Use Supplement

SECTION A. SCREENING FOR EVER/EVERY DAY/SOMEDAY SMOKING

A1 (Have/Has) (you/name) smoked at least 100 cigarettes in (your/his/her) entire life?

FR: 100 CIGARETTES = APPROXIMATELY 5 PACKS

(1) YES
(2) NO

BOX 0
IF A1 = 1, THEN GO TO A2
IF A1 = 2, D OR R AND RESPONDENT IS BETWEEN AGE 18 AND 34, THEN GO TO A3
ELSE, IF A1 = 2, D OR R, THEN GO TO SECTION J
This file documentation consists of the following materials:

<table>
<thead>
<tr>
<th>Attachment</th>
<th>Description</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>Abstract</td>
</tr>
<tr>
<td>2</td>
<td>Overview - Current Population Survey</td>
</tr>
<tr>
<td>3</td>
<td>Overview – May 2015 Tobacco Use Supplement</td>
</tr>
<tr>
<td>4</td>
<td>Glossary</td>
</tr>
<tr>
<td>5</td>
<td>How to Use the Record Layout</td>
</tr>
<tr>
<td>6</td>
<td>Basic CPS Record Layout</td>
</tr>
</tbody>
</table>
Variables already integrated in the harmonized dataset:
- Family income
- Education
- State
- Metropolitan status
- Race/ethnicity
- Country of birth (respondent, mother, father)
- Etc.
Pooling data across survey waves

Journal of Immigrant and Minority Health
https://doi.org/10.1007/s10903-019-00865-x

BRIEF COMMUNICATION

Heterogeneity in Tobacco-Use Behaviors Among U.S. Blacks per Global Region of Origin

Launick Saint-Fort1,2 · Kelvin Choi1

https://www.dreamstime.com/illustration/pooling-resources.html
Background

- 3.8 million of US Black individuals are foreign born (8.7% of the US Black population in 2013).
- These foreign-born Black individuals came from a diverse list of countries.
- Tobacco use behaviors may vary within the US Black population by global region of origin.
- Aim: To explore heterogeneity in tobacco use behaviors among US Black individuals by global region of origin.
Data

- Survey waves:
  - Prior waves were not used because of inconsistencies in race/ethnic definition and menthol cigarette smoking.

- Race: Black only
Global Region of Origin:

- US (Guam, Puerto Rico, the US Virgin Islands, and other US island areas; n=43,560)
- Africa (North and Sub-Saharan Africa; n=1,911)
- West Indies (n=2,194)
- Europe (n=192)
- Other regions were excluded (n=1,033)
Tobacco Use Measures

- Current cigarette smoking: At least some days
- Current cigar smoking: At least some days
- Established smokers: >100 cigarettes in a lifetime
  - Former smokers: Not currently smoking
  - Starting regular smoking as minor: Report age of first smoking regularly <18 years
- Current smokers:
  - Time to first cigarettes: <30 minutes after waking
  - Menthol: Regular use
Statistical Analysis

- Weighted
- Multivariable logistic regression model
  - Each tobacco use measure as a separate outcome
  - Adjusted for age, sex, education, household income, US census region, survey wave
Results: Current Tobacco Use

- Cigarettes:
  - US: 17.4%
  - Europe: 17.7%
  - Africa: 4.7%
  - West Indies: 4.9%
  - AOR = 0.25 (0.19, 0.33)

- Cigars:
  - US: 3.2%
  - Europe: 2.6%
  - Africa: 0.1%
  - West Indies: 1.1%
  - AOR = 0.41 (0.24, 0.70)

- Cigarettes and Cigars Combined:
  - US: 4.9%
  - Europe: 3.0%
  - Africa: 2.1%
  - West Indies: 1.1%
  - AOR = 0.24 (0.19, 0.30)
Results: Established Smokers

AOR=1.63 (1.19, 2.23)

40.0% US
33.1% Europe
51.0% Africa
55.3% West Indies

AOR=1.63 (1.10, 2.41)

45.5% US
42.1% Europe
32.7% Africa
45.4% West Indies

AOR=0.57 (0.38, 0.86)

Quit smoking

Started smoking as minor

NIH National Institute on Minority Health and Health Disparities
### Results: Current Smokers

<table>
<thead>
<tr>
<th>Region</th>
<th>First Cigarettes</th>
<th>Menthol</th>
</tr>
</thead>
<tbody>
<tr>
<td>US</td>
<td>47.4%</td>
<td>74.4%</td>
</tr>
<tr>
<td>Europe</td>
<td>42.1%</td>
<td>63.3%</td>
</tr>
<tr>
<td>Africa</td>
<td>25.5%</td>
<td>35.1%</td>
</tr>
<tr>
<td>West Indies</td>
<td>34.2%</td>
<td>54.0%</td>
</tr>
</tbody>
</table>

AOR = 0.47 (0.29, 0.76) for first cigarettes.
AOR = 0.19 (0.11, 0.34) for regular menthol.
Conclusions

• Not all Black individuals are the same!
• Heterogeneity in tobacco use behaviors exists by global region of origin.
• TUS-CPS can be a powerful data source to pool data together for otherwise difficult analyses!

This Photo by Unknown Author is licensed under CC BY-SA
Connect With Us

Visit us online www.nimhd.nih.gov

Connect with us on Facebook
www.facebook.com/NIMHD

Follow us on Twitter
@NIMHD

Join us on linkedin.com/company/nimhd-nih/

Sign up for news

Kelvin.choi@nih.gov
RAISE YOUR HAND if you wish to be unmuted and ask any final questions.

Ensure the Participants Panel is open.

Raise your hand by clicking on the hand icon.

Lower your hand by clicking on the hand icon again.
THANK YOU FOR YOUR PARTICIPATION
WE VALUE YOUR FEEDBACK!

Please share your feedback via a brief survey. The survey link will be shared via the Chat Box and email.

FOR MORE INFORMATION & HELPFUL RESOURCES

TUS-CPS Website
cancercontrol.cancer.gov/tus-cps

TUS-CPS Email Subscription
cancercontrol.cancer.gov/brp/tcrb/tus-cps#is-newsletter-subscription

TUS-CPS Team Contact
ncidccpsbrpadvances@mail.nih.gov